# Maintainer: Felix Yan <felixonmars@archlinux.org>
# Contributor: Daniel Nagy <danielnagy at gmx de>
# Contributor: Daniel Micay <danielmicay@gmail.com>

_hkgname=path-io
pkgname=haskell-path-io
pkgver=1.8.1
pkgrel=1
pkgdesc="Interface to ‘directory’ package for users of ‘path’"
url="https://github.com/mrkkrp/path-io"
license=("BSD")
arch=('loong64' 'x86_64')
depends=('ghc-libs' 'haskell-dlist' 'haskell-path' 'haskell-temporary'
         'haskell-unix-compat')
makedepends=('ghc' 'haskell-hspec')
source=("https://hackage.haskell.org/packages/archive/$_hkgname/$pkgver/$_hkgname-${pkgver}.tar.gz")
sha512sums=('9aa3a596a03e472c90c23356ae26b5442eb7e8cb145ccdf5ee2f4ef1d4ba7ab2a977f28e6e8605a983d64cad96061fd63276f01a9401919a8221d371376044c3')

build() {
  cd "$srcdir/$_hkgname-$pkgver"
  runhaskell Setup configure -O --enable-shared --enable-executable-dynamic --disable-library-vanilla \
        --prefix=/usr --docdir=/usr/share/doc/$pkgname --datasubdir=$pkgname --enable-tests \
        --dynlibdir=/usr/lib --libsubdir=\$compiler/site-local/\$pkgid --ghc-option=-fllvm \
            -f-dev
  runhaskell Setup build $MAKEFLAGS
  runhaskell Setup register --gen-script
  runhaskell Setup unregister --gen-script
  sed -i -r -e "s|ghc-pkg.*update[^ ]* |&'--force' |" register.sh
  sed -i -r -e "s|ghc-pkg.*unregister[^ ]* |&'--force' |" unregister.sh
}

check() {
  cd $_hkgname-$pkgver
  runhaskell Setup test --show-details=direct
}

package() {
  cd $_hkgname-$pkgver
  
  install -D -m744 register.sh "$pkgdir"/usr/share/haskell/register/$pkgname.sh
  install -D -m744 unregister.sh "$pkgdir"/usr/share/haskell/unregister/$pkgname.sh
  runhaskell Setup copy --destdir="$pkgdir"
  install -D -m644 LICENSE.md "$pkgdir"/usr/share/licenses/$pkgname/LICENSE.md
  rm -f "$pkgdir"/usr/share/doc/$pkgname/LICENSE.md
}
